
SUMMARY

Bifurcations of the periodic stationary solutions of nonlinear time-periodic time-delay dynamical systems
are analyzed. The solution operator of the governing nonlinear delay-differential equation is approximated
by a sequence of nonlinear maps via semidiscretization. The subsequent nonlinear maps are combined to
a single resultant nonlinear map that describes the evolution over the time period. Fold, flip, and Neimark-
Sacker bifurcations related to the fixed point of this map are analyzed via center manifold reduction and
normal form theorems. The analysis unfolds the approximate stability properties and bifurcations of the
stationary solution of the delay-differential equation, while it also allows the approximate computation of
the arising period-one, period-two, and quasi-periodic solution branches. The method is demonstrated for
the delayed Mathieu-Duffing equation and the results are verified by numerical continuation.

1. INTRODUCTION

Bifurcation analysis of time-delay systems has been a frequently discussed research topic in the
past few years. Several analytical approaches have been developed to investigate nonlinear delay-
differential equations (DDEs), such as the center manifold reduction [1, 2, 3] and the method
of multiple scales [4]. Due to the algebraic complexity of these methods, numerical approaches
for stability and bifurcation analysis of DDEs have also gained ground. The most well-known
computational tool is DDE-BIFTOOL [5], which enables the continuation of bifurcations and
branches of solutions in autonomous systems. DDE-BIFTOOL is also capable of computing critical
normal form coefficients for different kinds of codimension-1 and codimension-2 bifurcations in
autonomous DDEs. Other methods also allow numerical continuation for autonomous systems, see
e.g. [6], however, only a few techniques are dedicated to the analysis of time-periodic DDEs. The
package KNUT [7, 8] enables the continuation of bifurcations related to periodic solutions of time-
periodic DDEs. However, the computation of critical normal form coefficients associated with these
bifurcations is not implemented. To the best knowledge of the authors, so far the normal form

analysis realized in DDE-BIFTOOL for autonomous DDEs has not been extended to time-periodic
systems.

Here, we carry out approximate normal form analysis for time-periodic DDEs, by which the
stability and bifurcations of the stationary periodic solution can be investigated. Our approach is
based on discretizing the solution operator of the nonlinear time-periodic DDE using a nonlinear
map that describes the evolution over the time period. The fixed point of this map corresponds
to the stationary solution of the original DDE. Fold, flip, and Neimark-Sacker bifurcations of the
fixed point are associated with cyclic fold, period doubling, and secondary Hopf bifurcations of
the stationary solution, respectively, which may give rise to period-one, period-two, and quasi-
periodic solutions. We determine the stability of these solutions by analyzing the fixed point of
the corresponding nonlinear map and calculating its critical normal form coefficients. Via these
coefficients, we use analytical formulas to obtain the approximate amplitude of the bifurcating
solutions as a function of the bifurcation parameter. Therefore, as opposed to KNUT, this method
does not require the point-by-point continuation of the arising solutions, however, the results are
accurate in the vicinity of the bifurcation point only and secondary bifurcations cannot be detected.

The first step of the analysis is to discretize the solution operator of the nonlinear time-periodic
DDE. Several techniques exist for discretizing DDEs, see references [9, 10, 11] where some relevant
approaches are collected. The most popular and most efficient numerical methods include the
pseudospectral collocation [12, 13], the Chebyshev spectral continuous-time approximation [14],
the spectral element method [15], the spectral Legendre tau method [16], and the pseudospectral
tau method [17]. In what follows, we use the semidiscretization technique [18] to discretize
the solution operator of the DDE. This method formulates a sequence of nonlinear maps that
approximate the dynamics over the time period. Note that the approach of this paper is not restricted
to semidiscretization, it supports other discretization techniques as well, as long as the solution
operator is approximated by a (sequence of) nonlinear map(s) over the time period.

We show an algorithm to build a single resultant map from the sequence of nonlinear maps that
is correct up to third order in terms of the state variables. The bifurcation analysis of the resultant
map is performed according to the bifurcation theory of discrete maps discussed in [19, 20]. In [20],
center manifold reduction is used to reduce the dimension of maps in the vicinity of the bifurcation
point. Thereby, closed-form formulas are given in [20] for the critical normal form coefficients of
maps where the fixed point undergoes fold, flip, or Neimark-Sacker bifurcation. These normal form
coefficients can be used to analyze the bifurcation scenario in the original time-periodic DDE, and to
determine the stability and the approximate amplitude of solutions arising from cyclic fold, period
doubling, and secondary Hopf bifurcations, respectively.

The rest of the paper is organized as follows. Section 2 presents the semidiscretization for
nonlinear time-periodic DDEs. This results in a sequence of maps that is combined to a single
resultant nonlinear map in Sec. 3. Bifurcation analysis of the resultant map is presented Sec. 4, where
the normal form coefficients of fold, flip, and Neimark-Sacker bifurcations are given. Section 5
demonstrates the analysis of the delayed Mathieu-Duffing equation as a representative example,
and conclusions are drawn in Sec. 6.

2. THE TIME-PERIODIC SYSTEM AND ITS DISCRETIZATION

2.1. Governing Equation

In this paper, we investigate nonlinear time-periodic time-delay systems of form

ẏ(t) = f(t, y(t), y(t− τ)) , (1)

where y ∈ Rn. The function f : R+ ×Rn ×Rn → Rn is smooth in its second and third arguments
and periodic in its first argument, i.e., f(t, . , . ) = f(t+ T, . , . ) where T is the time period. We
assume that Eq. (1) has a time-periodic (stationary) solution yp(t) = yp(t+ T ) and we are interested
in the bifurcation of this solution. Therefore, we decompose the solution y(t) into the sum of the
stationary solution yp(t) and a perturbation u(t) as y(t) = yp(t) + u(t). We can write the variational

system corresponding to Eq. (1) in the form [21]

u̇(t) = D(t)u(t) + E(t)u(t− τ) + g(t,u(t),u(t− τ)) , (2)

where D(t),E(t) ∈ Rn×n, D(t) = D(t+ T ), E(t) = E(t+ T ) are the coefficient matrices of the
actual and the retarded linear terms, respectively, and g : R+ ×Rn ×Rn → Rn represents the
nonlinearities such that g(t, . , . ) = g(t+ T, . , . ) and g(t, 0, 0) ≡ 0. Note that u(t) ≡ 0 is the trivial
solution of Eq. (2), and the bifurcation analysis of yp(t) is equivalent to the analysis of the trivial
solution u(t) ≡ 0.

2.2. Semidiscretization

Here, we conduct the bifurcation analysis numerically via the semidiscretization technique [18].
By the discretization of the solution operator, semidiscretization resolves the time period T into
p ∈ N+ steps and considers the flow between the time instants tk = k∆t where k = 0, 1, ..., p and
∆t = T/p. The time-periodic terms are approximated in this case by constants during each time
step, thus Eq. (2) is approximated by

u̇(t) = Dku(t) + Eku(t− τ) + gk(u(t),u(t− τ)) , t ∈ [tk, tk+1) , (3)

with

Dk =
1

∆t

∫ tk+1

tk

D(t)dt , Ek =
1

∆t

∫ tk+1

tk

E(t)dt , gk( . , . ) =
1

∆t

∫ tk+1

tk

g(t, . , . )dt . (4)

Consider first the zeroth-order semidiscretization. The nonlinear and the time-delay terms are also
approximated by constants over the discretization interval [tk, tk+1):

u̇(t) = Dku(t) + Ekuk−r + gk(uk,uk−r) , [tk, tk+1) , (5)

where uk = u(tk) and r = bτ/∆tc. The semidiscretized system (5) can be solved over [tk, tk+1),
which yields

uk+1 = Pkuk + Rkuk−r + hk(uk,uk−r) (6)

where

Pk = eDk∆t , Rk =

∫ ∆t

0

eDk(∆t−s)ds Ek , hk( . , . ) =

∫ ∆t

0

eDk(∆t−s)ds gk( . , . ) . (7)

If D−1
k exists, then the integrals simplify to

Rk = (Pk − I) D−1
k Ek , hk( . , . ) = (Pk − I) D−1

k gk( . , . ) . (8)

In what follows, we employ the first-order semidiscretization to the linear delayed term in Eq. (3),
which gives

uk+1 = Pkuk + Rk,0uk−r + Rk,1uk−r+1 + hk(uk,uk−r) , (9)

where the coefficient matrices are given by

Rk,0 =

∫ ∆t

0

τ − (r − 1)∆t− s
∆t

eDk(∆t−s)ds Ek ,

Rk,1 =

∫ ∆t

0

s− τ + r∆t

∆t
eDk(∆t−s)ds Ek .

(10)

and can be simplified to

Rk,0 =

(
D−1
k +

1

∆t

(
D−2
k − (τ − (r − 1)∆t)D−1

k

)
(I− Pk)

)
Ek ,

Rk,1 =

(
−D−1

k +
1

∆t

(
−D−2

k + (τ − r∆t)D−1
k

)
(I− Pk)

)
Ek .

(11)

if D−1
k exists (see [18]).

Finally, map (9) can be represented in the form

zk+1 = Jkzk + Hk(zk) , (12)

with

zk =


uk

uk−1

...
uk−r

 , Jk =


Pk 0 · · · 0 Rk,1 Rk,0

I · · · 0 0
...

. . .
...

...
0 · · · I 0

 , Hk(zk) =


hk(uk,uk−r)

0
...
0

 , (13)

where I and 0 denote identity and zero matrices, respectively. Note that Jkzk represents the linear
term and Hk(zk) is purely nonlinear satisfying Hk(0) = 0, which implies that z = 0 is a fixed point
of Eq. (13).

3. GOVERNING NONLINEAR MAP

Map (12) approximates the evolution of system (2) over the time interval [tk, tk+1). Applying
map (12) successively p times, we get the resultant map z0 → zp in the form

zp = F(z0) , (14)

which represents the approximate evolution of system (2) over the whole period T . Therefore, we
carry out the (approximate) bifurcation analysis of the periodic solution yp(t) by deriving map (14)
and analyzing its fixed point z = 0. This section is dedicated to the derivation of the approximation
of map (14) that is correct up to third-order in terms of z0 and is suitable for bifurcation analysis.
The third-order approximation is of the form

zp = Az0 +
1

2
〈B, z0, z0〉+

1

6
〈C, z0, z0, z0〉+O

(
‖z0‖4

)
, (15)

where A, B, and C are second-, third-, and fourth-order coefficient matrices of the linear, quadratic,
and cubic terms, respectively, that are yet to be derived. Using index notation, these matrices read

A = [Ajl] =
[
∂lFj |0

]
, B = [Bjlm] =

[
∂l∂mFj |0

]
, C = [Cjlmq] =

[
∂l∂m∂qFj |0

]
, (16)

where ∂l represents the partial derivative with respect to the l-th element of z0 and subscript 0 stands
for the substitution z0 = 0. Whereas operations 〈 . , . 〉 and 〈 . , . , . 〉 are defined as

〈B, x, y〉 := [Bjlmxlym] , 〈C, x, y, z〉 := [Cjlmqxlymzq] . (17)

Note that these operations are linear and satisfy the following properties

〈K, x + y,u + v〉 =〈K, x,u〉+ 〈K, x, v〉+ 〈K, y,u〉+ 〈K, y, v〉 ,
〈K,Ux,Vy〉 =〈〈K,U,V〉, x, y〉 ,

U〈K, x, y〉 =〈UK, x, y〉 ,
〈K,Ux, 〈S, y, z〉〉 =〈〈K,U,S〉, x, y, z〉 ,
〈K, 〈S, x, y〉,Uz〉 =〈〈K,S,U〉, x, y, z〉 ,

〈L, x + y, z + u, v + w〉 =〈L, x, z, v〉+ 〈L, x, z,w〉+ 〈L, x,u, v〉+ 〈L, x,u,w〉
+ 〈L, y, z, v〉+ 〈L, y, z,w〉+ 〈L, y,u, v〉+ 〈L, y,u,w〉 ,

〈L,Ux,Vy,Wz〉 =〈〈L,U,V,W〉, x, y, z〉 ,
U〈L, x, y, z〉 =〈UL, x, y, z〉 ,

(18)

where U, V, and W are second-order matrices, K and S are third-order matrices, and L is a fourth-
order matrix. The corresponding operations with matrices are defined as

〈K,U,V〉 :=[KjlmUlqVms] , 〈K,U,S〉 := [KjlmUlqSmst] ,

〈K,S,U〉 :=[KjlmSlqsUmt] , 〈L,U,V,W〉 := [LjlmqUlsVmtWqx] .
(19)

This means that multiplications denoted by angle brackets are carried out with respect to the last
indices of the first operand and the first index of the other operands. These notations will be exploited
along this section.

Now we derive the coefficient matrices A, B, and C in Eq. (15) from map (12). Similarly to
Eq. (15), map (12) can be represented in the form

zk+1 = Jkzk +
1

2
〈Kk, zk, zk〉+

1

6
〈Lk, zk, zk, zk〉+O

(
‖zk‖4

)
. (20)

Matrices Kk, and Lk are defined by

K = [Kjlm] =
[
∂l∂mHj |0

]
, L = [Ljlmq] =

[
∂l∂m∂qHj |0

]
, (21)

where the semidiscretization index k was dropped in order to avoid confusion with the indices of the
matrices. Let us derive z1, z2, ..., zp as a function of z0 using map (20) successively, while dropping
all the terms of O

(
‖z0‖4

)
at each step. For k = 0 we get

z1 = J0z0 +
1

2
〈K0, z0, z0〉+

1

6
〈L0, z0, z0, z0〉+O

(
‖z0‖4

)
, (22)

where z1 is already expressed using z0. For k = 1 we obtain

z2 = J1z1 +
1

2
〈K1, z1, z1〉+

1

6
〈L1, z1, z1, z1〉+O

(
‖z1‖4

)
, (23)

where z1 should be substituted from Eq. (22) in order to get z2 as a function of z0. When substituting
Eq. (22), we drop the terms of O

(
‖z0‖4

)
and use the properties (18), by which we obtain

z2 =J1J0z0 +
1

2
J1〈K0, z0, z0〉+

1

6
J1〈L0, z0, z0, z0〉

+
1

2
〈〈K1, J0, J0〉, z0, z0〉+

1

4
〈〈K1, J0,K0〉, z0, z0, z0〉

+
1

4
〈〈K1,K0, J0〉, z0, z0, z0〉+

1

6
〈〈L1, J0, J0, J0〉, z0, z0, z0〉+O

(
‖z0‖4

)
.

(24)

In a similar manner, we can obtain z3, z4, ..., zp by recursively using Eq. (20). The final form of zp
becomes

zp = Az0 +
1

2
〈B̃, z0, z0〉+

1

6
〈C̃, z0, z0, z0〉+O

(
‖z0‖4

)
, (25)

where the coefficient matrices are

A =Qp−1,0 ,

B̃ =

p−1∑
j=0

Qp−1,j+1〈Kj ,Qj−1,0,Qj−1,0〉 ,

C̃ =

p−1∑
j=0

Qp−1,j+1〈Lj ,Qj−1,0,Qj−1,0,Qj−1,0〉

+
3

2

p−1∑
j=1

Qp−1,j+1

〈
Kj ,Qj−1,0,

j−1∑
l=0

Qj−1,l+1〈Kl,Ql−1,0,Ql−1,0〉

〉

+
3

2

p−1∑
j=1

Qp−1,j+1

〈
Kj ,

j−1∑
l=0

Qj−1,l+1〈Kl,Ql−1,0,Ql−1,0〉,Qj−1,0

〉
(26)

with the following notations

Qj,j+1 :=I , Qj,l := JjJj−1 . . . Jl+1Jl , l ≤ j , j = 0, . . . , p− 1 . (27)

Note that Qj,l represents the linear dynamics from the l-th to the (j + 1)-st time instant, that is,
Qj−1,0 describes the linear dynamics from the 0-th to the j-th time instant, whereas Qp−1,j+1

represents the linear dynamics from the (j + 1)-st to the p-th time instant. Thus, the terms in B̃
in Eq. (26) can be interpreted such that the linear dynamics governs up to the j-th time instant
(Qj−1,0), where the quadratic term has its effect (Kj), and then the linear dynamics operates again
from the (j + 1)-st instant (Qp−1,j+1) – and the results for all possible values of j are added. The
cubic term with C̃ is similarly constructed from the effect of two quadratic terms (Kj and Kl) or a
single cubic term (Lj) and linear maps at all other time instants. Since we do not use the terms of
O
(
‖z0‖4

)
, no other combination of nonlinear terms needs to be considered.

Finally, note that matrices B, C in Eqs. (15)-(16) and B̃, C̃ in Eqs. (25)-(26) are not necessarily
the same – the coefficient matrices of the nonlinear terms are nonunique as different B and C
matrices can produce the same expressions for 〈B, z0, z0〉 and 〈C, z0, z0, z0〉, respectively. However,
the matrices B and C given by Eq. (16) originate in second and third derivatives, thus they are
symmetric to their last two and three indices, respectively: Bjlm = Bjml and Cjlmq = Cjlqm =
Cjmlq = Cjmql = Cjqlm = Cjqml. Therefore, it can be shown that B and C in Eq. (16) are the
symmetric parts of B̃ and C̃ in Eq. (26), respectively, which implies

B =sym(B̃) =
1

2
[B̃jlm + B̃jml] ,

C =sym(C̃) =
1

6
[C̃jlmq + C̃jlqm + C̃jmlq + C̃jmql + C̃jqlm + C̃jqml] .

(28)

Notice that according to Eq. (26), matrix B̃ is in fact symmetric and B = B̃, while C̃ is not symmetric
and C 6= C̃.

4. BIFURCATION ANALYSIS

In this section, we perform bifurcation analysis based on map (15). In order to emphasize
that this map describes the evolution over the principal period of length T = p∆t (i.e., over p
semidiscretization steps), we introduce a new index K ∈ N that denotes the number of principal
periods elapsed. Introducing ZK = zKp, map (15) can be rewritten as

ZK+1 = AZK +
1

2
〈B,ZK ,ZK〉+

1

6
〈C,ZK ,ZK ,ZK〉+O

(
‖ZK‖4

)
. (29)

Once the coefficient matrices A, B, and C are calculated from Eqs. (26)-(28), map (29) can be
used for bifurcation analysis. The bifurcation theory of nonlinear maps is discussed in [20] – from
this point on, we apply the formulas presented therein. Note that Neimark-Sacker, flip, and fold
bifurcations of the fixed point Z = 0 of map (29) correspond to secondary Hopf (torus), period
doubling, and cyclic fold bifurcations of the periodic orbit yp(t) of Eq. (1). Here, we demonstrate
the analysis of Neimark-Sacker bifurcation in detail and we address the differences for flip and fold
bifurcations.

4.1. Neimark-Sacker Bifurcation

Let α ∈ R denote the bifurcation parameter and let αcr be the Neimark-Sacker bifurcation point.
The bifurcation of the fixed point is determined by the eigenvalues of the coefficient matrix A,
which are given by the characteristic equation det(µI−A) = 0. At Neimark-Sacker bifurcation, a
pair of complex conjugate eigenvalues is located on the unit circle of the complex plane, that is,
there exists a critical eigenvalue

µcr = eiθ0 , |µ|cr = 1 , (30)

Copyright c© 2017 John Wiley & Sons, Ltd. Int. J. Numer. Meth. Engng (2017)

BIFURCATION ANALYSIS OF NONLINEAR TIME-PERIODIC TIME-DELAY SYSTEMS 7

where i2 = −1, θ0 ∈ (0, π). From this point on, subscript cr refers to the substitution µ = µcr,
α = αcr. We can determine the left and right eigenvectors p and q of the linear coefficient matrix
Acr associated with the critical eigenvalue µcr by solving

Acrq = µcrq , pAcr = µcrp , pq = 1 , (31)

where over-bar indicates complex conjugate.
The analysis of the Neimark-Sacker bifurcation can be done via center manifold reduction [20].

Accordingly, the dynamics of map (29) is restricted to a two-dimensional center manifold spanned
by (the real and imaginary parts of) q and q. The approximation of the two-dimensional restricted
map can be written in polar form as

ρ̂K+1 =ρ̂K + |µ|′cr(α− αcr)ρ̂K + acrρ̂
3
K +O

(
ρ̂4
K

)
,

θ̂K+1 =θ̂K + θ0 +O
(
ρ̂2
K

)
,

(32)

where ρ̂ is the amplitude and θ̂ is the phase angle. Note that the coefficients of ρ̂K , ρ̂3
K , and

θ̂0
K = 1 are approximated by linear and constant functions of the bifurcation parameter α – this

approximation is valid only in the vicinity of the bifurcation (α ≈ αcr). Prime indicates derivative
with respect to the bifurcation parameter α, |µ|′cr is the root tendency (the radial speed by which the
critical eigenvalues cross the unit circle), and acr is the leading coefficient for which a closed-form
formula is available in [20], see below.

In Eq. (32), the map for the amplitude ρ̂ has a nontrivial fixed point

ρ =

√
−|µ|

′
cr(α− αcr)

acr
+O(α) . (33)

On the two-dimensional center manifold, this nontrivial fixed point corresponds to a critical solution
that is located on an isolated closed invariant curve with radius ρ. The genericity conditions related to
the existence and uniqueness of the closed invariant curve are the transversality condition |µ|′cr 6= 0
and the nondegeneracy conditions µmcr 6= 1, m = 1, 2, 3, 4, acr 6= 0, see [20]. Here, we assume that
these conditions are satisfied. The critical solution associated with the closed invariant curve is

ZK = ρeiKθ0q + ρe−iKθ0q . (34)

In order to obtain solution (34), the root tendency |µ|′cr and the leading coefficient acr must be
determined. The calculation of the root tendency |µ|′cr can be reduced to determining the constant
µ′cr and using the relationship [22]

|µ|′ =
1

|µ|
Re (µµ′) (35)

for the critical case µ = µcr. The constant µ′cr can be obtained by the implicit differentiation of the
characteristic equation det(µI−A) = 0:

µ′cr =
dµ

dα

∣∣∣∣
cr

= −

∂

∂α
det(µI−A)

∂

∂µ
det(µI−A)

∣∣∣∣∣∣∣
cr

= − tr(adj(µcrI−Acr)(−A′cr))

tr(adj(µcrI−Acr))
. (36)

Note that the derivative A′ can be given using the derivative J′j of the linear coefficient matrices of
each semidiscretization step:

A′ =

p−1∑
j=0

Qp−1,j+1J′jQj−1,0 . (37)

As for the leading coefficient acr, formula (5.74) of [20] can directly be used:

acr =
1

2
Re
(
e−iθ0

(
p〈Ccr,q,q,q〉+ 2p〈Bcr,q, (I−Acr)

−1〈Bcr,q,q〉〉

+p〈Bcr,q, (e2iθ0I−Acr)
−1〈Bcr,q,q〉〉

))
. (38)

Notice that the magnitude (norm) of the eigenvector q is not determined by Eq. (31), it can be chosen
arbitrarily. When multiplying the eigenvector q by a constant c ∈ C, the coefficient acr scales by |c|2,
the amplitude ρ scales by 1/|c|, while the critical solution (34) remains the same. Also note that if the
l-th component of q is selected to be ql = 1/2, then ρ represents the amplitude of l-th component of
solution (34) asZK,l = ρ cos(Kθ0). Otherwise, Eq. (34) impliesZK,l = 2|ql|ρ cos(Kθ0 + ϑl) where
ϑl is a certain phase shift and 2|ql|ρ represents the amplitude of l-th component of solution (34).

The critical solution (34) corresponds to a quasi-periodic solution yqp(t) of flow (1) that may
simplify to a periodic solution in special cases. The stability of the quasi-periodic solution is the
same as the stability of the closed invariant curve associated with Neimark-Sacker bifurcation,
which is determined by the sign of the leading coefficient acr. If acr < 0, the Neimark-Sacker
bifurcation is supercritical and solution (34) is stable, while if acr > 0, the bifurcation is subcritical
and the solution is unstable.

In addition, the fixed point ρ is related to the amplitude of the quasi-periodic solution yqp(t).
Suppose that we are interested in the half r of the peak-to-peak amplitude in terms of the j-th
coordinate yqp,j(t) of the vector yqp(t). Then, the amplitude rp of the periodic solution yp,j(t)
must be added to the amplitude ru of the perturbation uj(t) that can be determined from ρ as
follows. According to Eq. (13), the j-th, (j + n)-th, ..., (j + rn)-th elements of solution (34) are
uKp,j , uKp−1,j , ..., uKp−r,j . These discrete-time solutions vary with amplitudes ρ0, ρ1, ..., ρr,
where the largest of them approximates the amplitude of uj(t): ru = max(ρm), m = 1, 2, ..., r.
The amplitudes ρm can be obtained from Eq. (33) provided that the eigenvector q is chosen such
that qj+mn = 1/2. Equivalently, instead of determining ρ0, ρ1, ..., ρr, we can calculate the leading
coefficient acr from Eq. (38) using an eigenvector q with arbitrary norm, and then scale it as

ãcr =
acr

4 max
0≤m≤r

(
|qj+mn|2

) , (39)

This way, Eq. (33) gives ru directly:

ru =

√
−|µ|

′
cr(α− αcr)

ãcr
. (40)

In summary, the amplitude r = rp + ru and the stability of the quasi-periodic solution can be
determined by Eqs. (35)-(40), where the formula of the coefficient matrices Acr, Bcr, and Ccr are
given by Eqs. (26)-(28).

4.2. Flip Bifurcation

The analysis of flip bifurcation is similar to that presented for Neimark-Sacker bifurcation. At flip
bifurcation, the critical eigenvalue µcr = −1 is located on the unit circle of the complex plane. The
corresponding eigenvectors p and q are real, and q spans a one-dimensional center manifold. The
restriction of map (29) to this manifold has the equivalent form

ρ̂K+1 = −ρ̂K − µ′cr(α− αcr)ρ̂K − acrρ̂
3
K +O

(
ρ̂4
K

)
(41)

provided that the transversality condition µ′cr 6= 0 and the nondegeneracy condition acr 6= 0 are
fulfilled [20].

The second iterate of map (41) has a nontrivial fixed point ρ given by Eq. (33). This fixed point
corresponds to a period-two solution of map (29) that approximately reads

ZK = ρ(−1)Kq (42)

and is associated with a period-two solution of flow (1). The stability of the period-two solution is
determined by the sign of acr (it is stable for acr < 0 and unstable for acr > 0), while its approximate
amplitude r = rp + ru can be obtained the same way as that of the quasi-periodic orbit in the
previous section. Namely, the constant µ′cr is real (|µ|′cr = µ′cr) and can be calculated according

Figure 1. Stability chart of the delayed Mathieu-Duffing equation (49) for a1 = 0.1, ε = 1, τ = 2π. Vertical
lines at δ = 2, 0.7, and 0.9 correspond to the bifurcation diagrams shown in Figs. 2, 3, and 4, respectively.

to Eq. (36). Whereas the expression of the leading coefficient modifies to

acr = −1

6
p〈Ccr,q,q,q〉+

1

2
p〈Bcr,q, (Acr − I)−1〈Bcr,q,q〉〉 , (43)

see the formulas after Eq. (5.69) in [20]. The following scaled coefficient can be used to obtain the
amplitude of the period-two solution from Eq. (40):

ãcr =
acr

max
0≤m≤r

(
q2
j+mn

) . (44)

4.3. Fold Bifurcation

In the case of fold bifurcation, the critical eigenvalue is µcr = 1, the eigenvectors p and q are
real, and q spans a one-dimensional center manifold. Restriction to this manifold yields the critical
system [20]

ρ̂K+1 = ρ̂K + µ′cr(α− αcr)ρ̂K + σcrρ̂
2
K + acrρ̂

3
K +O

(
ρ̂4
K

)
. (45)

According to the formulas after Eq. (5.68) in [20], the critical normal form coefficients read

σcr =
1

2
p〈Bcr,q,q〉 , (46)

acr =
1

6
p〈Ccr,q,q,q〉 −

1

2
p〈Bcr,q, (Acr − I)INVd〉 ,

d =〈Bcr,q,q〉 − (p〈Bcr,q,q〉) q ,
(47)

where (Acr − I)INVd is obtained by solving the following equation for w:[
Acr − I q

p 0

][
w
v

]
=

[
d
0

]
. (48)

Note that since µcr = 1 is an eigenvalue of Acr, the matrix Acr − I is singular. Thus, its inverse does
not exist and instead we use (Acr − I)INV as defined above. The nondegeneracy condition associated
with cyclic fold bifurcation is σcr 6= 0 [20].

5. A CASE STUDY: THE DELAYED MATHIEU-DUFFING EQUATION

In this section, we demonstrate the bifurcation analysis via semidiscretization for the delayed
Mathieu-Duffing equation:

ẍ(t) + a1ẋ(t) + (δ + ε cos t)x(t) + µx3(t) = bx(t− τ) . (49)

Copyright c© 2017 John Wiley & Sons, Ltd. Int. J. Numer. Meth. Engng (2017)

Figure 2. Bifurcation diagrams of the delayed Mathieu-Duffing equation near a secondary Hopf bifurcation
assuming case A of Fig. 1 with µ = 0.5, δ = 2 (a); the convergence of the normal form coefficients (b); the

error relative to the solution of KNUT (c).

This equation was analyzed by the method of averaging in [23] for the undamped case (a1 = 0).
Analyses of the Mathieu equation in the absence of delay (b = 0) can be found in [24, 25, 26]
considering cubic nonlinearities and in [27] for the case of quadratic damping. Other analytical
approaches for studying nonlinear delay-free time-periodic systems can also be found in [28, 29,
30, 31]. The closed-form stability condition and the stability diagram of the linear delayed Mathieu
equation (µ = 0) was given in [32].

Equation (49) defines a nonlinear time-periodic time-delay system that can be represented in the
form of Eq. (1). We analyze the stability and bifurcations of its periodic solution, the trivial solution
xp(t) ≡ 0 itself. We decompose the solution of Eq. (49) into the sum of the periodic solution and a
perturbation: x(t) = xp(t) + ξ(t) that yields Eq. (2) with

u(t) =

[
ξ(t)

ξ̇(t)

]
, D(t) =

[
0 1

−(δ + ε cos t) −a1

]
, E(t) ≡ E =

[
0 0
b 0

]
,

g(t,u(t),u(t− τ)) ≡ g(u(t)) =

[
0

−µξ3(t)

]
.

(50)

Figure 3. Bifurcation diagrams of the delayed Mathieu-Duffing equation near a period doubling bifurcation
assuming case B of Fig. 1 with µ = 0.5, δ = 0.7 (a); the convergence of the normal form coefficients (b);

the error relative to the solution of KNUT (c).

Its semidiscretized counterpart is given by Eq. (5) with

Dk =

 0 1

−
(
δ + ε

sin tk+1 − sin tk
∆t

)
−a1

 , gk(uk) =

[
0
−µx3

k

]
(51)

that can be written in form (12) using Eqs. (7), (11) and (13).
From this point on, the bifurcation analysis can be carried out according to Sec. 4. The results

are presented in Figs. 1-4 for a1 = 0.1, ε = 1, µ = 0.5, τ = 2π. Figure 1 presents the stability chart
of the system in the plane (δ, b) that was computed by the semidiscretization method using period
resolution p = 50. The parameter regions associated with linearly stable trivial solution are shown
by gray shading, whereas the loci of cyclic fold, period doubling, and secondary Hopf bifurcations
are also indicated. As shown by the vertical lines B, C, and A in Fig. 1, δ = 0.7, δ = 0.9, and
δ = 2 are selected for analyzing period doubling, cyclic fold, and secondary Hopf bifurcations,
respectively, and b is chosen as bifurcation parameter. The bifurcation points under analysis are
indicated by circles, while the corresponding bifurcation diagrams are shown in Figs. 2, 3, and 4.
The Mathematica code generating these figures is given in Appendix A.

Figure 4. Bifurcation diagrams of the delayed Mathieu-Duffing equation near a cyclic fold bifurcation
assuming case C of Fig. 1 with µ = 0.5, δ = 0.9 (a); the convergence of the normal form coefficients (b);

the error relative to the solution of KNUT (c).

For δ = 2, a secondary Hopf bifurcation occurs at b = bcr ≈ 0.5 that gives rise to a quasi-periodic
solution. The amplitude r of this solution as a function of the bifurcation parameter b is indicated
by red lines in Fig. 2(a), where semidiscretization with period resolutions p = 5, 10, ..., 50, 55
was used to obtain the bifurcation diagrams. The results were verified by numerical continuation:
thick black line shows the amplitude ‖x(t)‖∞ obtained by the continuation package KNUT [7, 8].
The bifurcation point (determined by the value of bcr) and the initial curvature (given by the ratio of
ãcr and |µ|′cr) of the bifurcation diagrams obtained by semidiscretization converge to those obtained
by KNUT. Note, however, that the analysis based on normal forms is valid only in the vicinity of
the bifurcation: it cannot catch phenomena such as the fold of the quasi-periodic orbit shown by the
thick black line. Figure 2(c) shows the error of semidiscretization with different period resolutions
relative to the results of KNUT. Notice that KNUT computes a branch of bifurcating solutions, while
it does not provide the normal form coefficients directly and only the values of bcr and ãcr/|µ|′cr are
determined by the branch. Therefore, we used the first point of the branch as bifurcation point bcr

and we calculated the ratio ãcr/|µ|′cr from the initial curvature of the branch via fitting a parabola
to its first 25 points and using Eq. (40). According to the figure, the approach of KNUT and the
method of this paper converge to the same result. The convergence of semidiscretization is also

illustrated in Fig. 2(b) where the values of bcr, |µ|′cr, and ãcr are shown as a function of the period
resolution p. Note that the sign of ãcr that determines the stability of the arising quasi-periodic orbit
is correct even for the smallest value of p. This implies that even a rough discretization is sufficient
to determine the sense (criticality) of a bifurcation.

Fig. 3 shows the bifurcation diagrams and the normal form coefficients of a period doubling
bifurcation occurring for δ = 0.7. The bifurcation gives rise to a stable period-two branch that is
shown by green lines for semidiscretization with period resolutions p = 5, 10, ..., 50, 55. Thick
black line shows the result of numerical continuation using KNUT. In this case, the bifurcation
diagrams obtained by the two methods almost overlap, the approximate amplitude obtained by
normal form analysis is valid for the depicted range of the bifurcation parameter. In addition, the
sign of the leading coefficient ãcr is again the same for all values of p, thus even a small period
resolution is suitable to analyze the criticality of the bifurcation.

Fig. 4 shows the case of a cyclic fold bifurcation for δ = 0.9. Note that the bifurcation is
nongeneric in this case, since the normal form coefficient σcr in Eq. (45) is zero as the coefficient
B of the quadratic terms is a zero matrix. The bifurcation associated with map (45) is therefore
a pitchfork bifurcation that gives rise to an unstable branch of a nontrivial periodic solution. The
amplitude of this solution can be obtained the same way as that of the period-two orbit associated
with period doubling bifurcation – via Eqs. (40) and (44). The results of semidiscretization (shown
by red lines) are again in agreement with those obtained by numerical continuation using KNUT
(shown by thick black line).

Finally, we make comparison with the results of [23] where the undamped case a1 = 0 was
investigated analytically using the method of averaging by assuming small ε, µ, b and considering
the vicinity of δ = 1/4. We consider Example 1 of [23] with parameters ε = 0.05, µ = 0.05,
b = 0.0125 (these correspond to α = 1, β = 0.25, γ = 1, ε = 0.05 using the notations of [23]).
Fig. 5(a) shows the stability chart of the associated linear system in the plane (τ, δ) for τ ∈ [0, 3π],
δ ∈ [0.2, 0.3] (these correspond to T ∈ [0, 3π], δ1 ∈ [−1, 1] using the notations of [23]). Here, the
semidiscretization method with period resolution p = 30 was used to compute the stability chart
– the corresponding figure obtained by averaging is shown in Fig. 10 of [23]. Fig. 5(b) shows the
bifurcation diagrams for τ = π that corresponds to the solid vertical line in Fig. 5(a). For τ = π,
the trivial solution x(t) ≡ 0 undergoes period doubling bifurcations when its stability changes
at δ ≈ 0.23 and δ ≈ 0.27. The bifurcations give rise to period-two orbits. In [23], the following

approximate analytical formula was derived for the amplitude of the arising orbits:

r =

√
4

3µ

(
b cos

(τ
2

)
−
(
δ − 1

4

)
± 1

2

√
ε2 − 4b2 sin2

(τ
2

))
, (52)

see Eq. (19) in [23]. The amplitude r obtained using formula (52) is shown by dashed lines in
Fig. 5(b) and the results of semidiscretization for period resolution p = 30 are shown by solid lines.
The approximate analytical results obtained by averaging agree well with the numerical results of
semidiscretization.

6. CONCLUSIONS

In this paper, we presented an approach to analyze bifurcations of periodic solutions associated with
time-periodic DDEs. The method uses normal form coefficients to unfold the possible bifurcation
scenarios and to determine the amplitude and stability of solutions arising from bifurcation. To this
end, we discretized the solution operator of the time-periodic DDE by a sequence of nonlinear maps
using the semidiscretization technique. Extension to systems with distributed and time-periodic time
delays is also possible via semidiscretization. Note, however, that other discretization schemes that
lead to a sequence of nonlinear maps could also be used, the analysis built on these maps would be
the same. The main contribution of this work is the algorithm to combine the subsequent maps into a
single resultant map that enables us to analyze bifurcations via center manifold theory. The analysis
of the fixed point reveals the behavior of the periodic solution related to the original time-periodic
DDE, as well as the stability and amplitude of the bifurcating orbits.

The computational costs of semidiscretization are determined by the resolution of the time
period and the delay. The delay resolution r affects memory consumption and the dimension of
eigenproblems to be solved in Eq. (31), since the coefficient matrices A, B, and C are of size
(r + 1)n. This dimension can be reduced for dynamical systems that do not involve time delays in
all state variables, since the corresponding delayed states can be omitted from vector zk in Eq. (13).
This is the case for the delayed Mathieu-Duffing equation (49), too, where ẋ does not show up
with retarded argument. The period resolution p affects the number of matrix multiplications in
Eqs. (26)-(27), which is 2p2 + 8p− 10. In systems where second-order terms are absent (such
as in the delayed Mathieu-Duffing equation), the number of matrix multiplications decreases to
(p2 + 7p− 8)/2, because the terms containing Kk drop. In addition, the sparsity of matrix Jk
in Eq. (13) can also be taken advantage of in order to carry out matrix multiplications with less
operations, see Sec. 3.4 in [18]. On the other hand, computational costs can be reduced further by
replacing semidiscretization with more effective discretization techniques [12, 13, 14, 15, 16, 17].
This changes Eq. (13) only, while the subsequent analysis remains the same. These techniques
operate with significantly smaller coefficient matrices and less matrix multiplications. Furthermore,
when multiple, distributed, or time-periodic delays occur, Eq. (13) should be modified only.

Apart from compatibility with other discretization techniques, the main advantage of the method
is that only three constant parameters have to be determined: the bifurcation point αcr, the root
tendency |µ|′cr, and the leading coefficient ãcr; it is not required to compute branches of solutions
point-by-point. Using the three constant parameters, it is possible to determine the approximate
amplitude of the emerging period-one, period-two, or quasi-periodic solutions, although the
amplitude is accurate in the vicinity of the bifurcation only. However, the sense (the criticality)
of the bifurcation, that is, the stability of the arising solutions can also be determined based on the
sign of the leading coefficient ãcr. According to the examples of this paper, the sign of ãcr can be
obtained even by a rough discretization of the solution operator, e.g. by using a period resolution
p = 5 – and this is typically the case when ãcr is not close to zero. This property makes the method
a fast tool for criticality analysis.

Analyzing the criticality of bifurcations is a relevant issue in engineering. From engineering point
of view, subcritical bifurcations are considered more dangerous than supercritical ones. Unstable
solutions arising from a subcritical bifurcation make the basin of attraction of the linearly stable

stationary solution finite. This leads to locally but not globally stable engineering systems where
oscillations may evolve due to large enough perturbations. This phenomenon is often referred to
as bistability. In engineering, bistability is dangerous and must be avoided, since the designed state
of the system is stable, but it may eventually lose stability to large enough external disturbances.
Our future research involves engineering applications of the method presented above: for instance,
the criticality of period doubling and secondary Hopf bifurcations related to milling operations can
be analyzed. The regenerative machine tool vibrations in milling are described by nonlinear time-
periodic DDEs, for which the phenomenon of bistability was shown to occur [33].

ACKNOWLEDGEMENTS

This work has received funding from the János Bolyai Research Scholarship of MTA (BO/00589/13/6)

A. MATHEMATICA CODE

Quit;

(*PARAMETERS*)

a1=0.1; delta=0.7; epsi=1.; tau=2.*Pi; mu=0.5;
(*Select the bifurcation parameter
and give its estimation at the bifurcation*)
b0=par; parst=-0.2;

(*PERIOD AND SEMIDISCRETIZATION STEP*)

(*time period*)
T=2.*Pi;
(*period resolution*)
p=10;
(*time step*)
dt=T/p;
(*delay resolution*)
r=IntegerPart[Chop[tau/dt]];

(*SYSTEM DEFINITION*)

(*state vector*)
u[i_]={x[i],xdot[i]};
(*system dimension*)
n=Length[u[i]];
(*determine amplitude of orbits in terms of the j-th state*)
jj=1;
(*set of linear coefficient matrices*)
DD[par_]=Table[{{0,1},
{-(delta+epsi*(Sin[(i+1)*dt]-Sin[i*dt])/dt),-a1}},{i,0,p-1}];
(*set of retarded coefficient matrices*)
EE[par_]=Table[{{0,0},{b0,0}},{i,0,p-1}];
(*set of nonlinear terms*)
g[par_]=Table[{0,-mu*Power[x[i],3]},{i,0,p-1}];

(*SEMIDISCRETIZED SYSTEM MATRICES*)

PP[par_]=MatrixExp[#*dt]&/@DD[par];
R0[par_]=MapThread[Dot,{Inverse/@DD[par],EE[par]}]+
1/dt*MapThread[Dot,{((Inverse[#].Inverse[#])&/@DD[par]-

(tau-(r-1)*dt)*Inverse/@DD[par]),
(IdentityMatrix[n]-#)&/@PP[par],EE[par]}];
R1[par_]=-MapThread[Dot,{Inverse/@DD[par],EE[par]}]+
1/dt*MapThread[Dot,{(-(Inverse[#].Inverse[#])&/@DD[par]+
(tau-r*dt)*Inverse/@DD[par]),
(IdentityMatrix[n]-#)&/@PP[par],EE[par]}];
h[par_]=MapThread[Dot,{(#-IdentityMatrix[2])&/@PP[par],
Inverse/@DD[par],g[par]}];
(*augmented state vector*)
z=Table[Flatten[Table[u[i-k],{k,0,r}]],{i,0,p-1}];
(*linear part of map*)
JJ[par_]=Table[Join[ArrayFlatten[{{PP[par][[i+1]],
Table[0,{n},{(r-2)*n}],R1[par][[i+1]],R0[par][[i+1]]}}],
ArrayFlatten[{{IdentityMatrix[r*n],Table[0,{r*n},{n}]}}]],{i,0,p-1}];
(*nonlinear part of map*)
H[par_]=MapThread[Join,{h[par],Table[0,{p},{r*n}]}];

(*BIFURCATION POINT*)

II=IdentityMatrix[Last[Dimensions[JJ[par]]]];
(*linear map as function of the bifurcation parameter*)
AApar[par_?NumberQ]:=Nest[{#[[1]]+1,Chop[JJ[par][[#[[1]]+1]].#[[2]]]}&,
{0,II},p][[2]];
(*corrected bifurcation point*)
parst=par/.FindRoot[Abs[Eigenvalues[AApar[par],1]]-1,
{par,parst}];
(*map at bifurcation point*)
JJst=Chop[JJ[parst]];
Hst=Chop[H[parst]];

(*DEFINITION OF <.,.,.> AND <.,.,.,.> PRODUCTS*)

(*<.,.,.> product between third and second order matrices*)
ThreeTwoProduct[X_,Y_,Z_]:=Transpose[Transpose[
X,{1,3,2}].Y,{1,3,2}].Z;
(*<.,.,.> product between third and third order matrices*)
ThreeThreeProduct[X_,Y_,Z_]:=Transpose[Transpose[
X,{1,3,2}].Y,{1,4,2,3}].Z;
(*<.,.,.,.> product between fourth and second order matrices*)
FourTwoProduct[X_,Y_,Z_,W_]:=Transpose[Transpose[Transpose[
X,{1,4,2,3}].Y,{1,4,2,3}].Z,{1,4,2,3}].W;

(*RESULTANT MAP*)

(*LINEAR PART*)
(*matrices Q_{j-1,0}*)
Mj=NestList[{#[[1]]+1,Chop[JJst[[#[[1]]+1]].#[[2]]]} &,
{0,II},p-1][[All,2]];
(*matrices Q_{p-1,j+1}*)
Nj=Reverse[NestList[{#[[1]]-1,Chop[#[[2]].JJst[[#[[1]]-1]]]} &,
{p+1,II},p-1][[All,2]]];
(*matrix A*)
AA=Chop[JJst[[p]].Mj[[p]]];

(*QUADRATIC PART*)
Kj=(D[#[[1]],{#[[2]]},{#[[2]]}]/.Thread[#[[2]]->0]) &/@
(Transpose[{Hst,z}]);
(*<K_j,Q_{j-1,0},Q_{j-1,0}>*)
KMMj=MapThread[ThreeTwoProduct,{Kj,Mj,Mj}];
(*matrix B*)
BB=Total[MapThread[Dot,{Nj,KMMj}]];

(*CUBIC PART*)
Lj=(D[#[[1]],{#[[2]]},{#[[2]]},{#[[2]]}]/.Thread[#[[2]]->0]) &/@

(Transpose[{Hst,z}]);
(*<L_j,Q_{j-1,0},Q_{j-1,0},Q_{j-1,0}>*)
LMMMj=MapThread[FourTwoProduct,{Lj,Mj,Mj,Mj}];
(*matrices Subscript[Q,j-1,l+1]*)
Qjl=Table[Reverse[NestList[{#[[1]]-1,Chop[#[[2]].JJst[[#[[1]]-1]]]} &,
{j+1,II},j-1][[All,2]]],{j,1,p-1}];
(*Sum(Q_{j-1,l+1}<K_l,Q_{l-1,0},Q_{l-1,0}>)*)
QKMMj=Join[{0.*KMMj[[1]]},Total[MapThread[
Dot,{#,KMMj[[1;;Length[#]]]}]]&/@Qjl];
KQKMj=(MapThread[ThreeThreeProduct,{Kj,QKMMj,Mj}]+
MapThread[ThreeTwoProduct,{Kj,Mj,QKMMj}])/2.;
(*matrix \tilde{C}*)
CCt=Total[MapThread[Dot,{Nj,LMMMj}]]+3.*Total[MapThread[Dot,{Nj,KQKMj}]];
(*matrix C*)
CC=1/6*Total[Transpose[CCt,#]&/@(Prepend[#,1]&/@Permutations[{2,3,4}])];

(*RELEVANT EIGENVALUES AND EIGENVECTORS*)

sec=SessionTime[];
eigs=Eigensystem[AA,2];
theta0=Arg[eigs[[1,1]]];
(*critical eigenvalue*)
mu0=Exp[I*theta0]//Chop;
(*critical right eigenvector*)
qq=eigs[[2,1]]//Chop;
eigsT=Eigensystem[Transpose[AA],2];
(*critical left eigenvector*)
pp=If[Im[mu0]==0,eigsT[[2,1]],eigsT[[2,2]]]//Chop;
(*normalization of eigenvectors*)
pp /= Conjugate[Conjugate[pp].qq]//Chop;
(*type of bifurcation*)
If[Im[mu0]==0,If[mu0>0,Print["Nongeneric fold (pitchfork) bifurcation"],
Print["Flip bifurcation"]],Print["Neimark-Sacker bifurcation"]]
StringJoin["Bifurcation point: ",ToString[parst]]

(*ROOT TENDENCY*)

sec=SessionTime[];
(*command for matrix adjoint*)
Adj[m_]:=Map[Reverse,Minors[Transpose[m],Length[m]-1],{0,1}]*
Table[Power[-1,i+j],{i,Length[m]},{j,Length[m]}];
(*A’cr*)
AAp0=Total[MapThread[Dot,{Nj,Derivative[1][JJ][parst],Mj}]];
(*mu’cr*)
mup0=-Tr[Adj[mu0*II-AA].(-AAp0)]/Tr[Adj[mu0*II-AA]];
(*|mu|’cr*)
s0=(Re[Conjugate[mu0]*mup0]/Abs[mu0])//Chop;
StringJoin["Root tendency: ",ToString[s0]]

(*LEADING COEFFICIENT*)

sec=SessionTime[];
(*leading coefficient acr*)
a00=If[Im[mu0]==0,If[mu0>0,
(*fold*)
1/6*pp.CC.qq.qq.qq-1/2*pp.BB.qq.
(LinearSolve[Append[Flatten/@Thread[{AA-II,qq}],Join[pp,{0}]],
Join[BB.qq.qq-(pp.BB.qq.qq) qq,{0}]][[1;;-2]]),
(*flip*)

-(1/6*pp.CC.qq.qq.qq-1/2*pp.BB.qq.Inverse[AA-II].BB.qq.qq)],
(*Neimark-Sacker*)
1/2*Re[Exp[-I*theta0]*(Conjugate[pp].CC.qq.qq.Conjugate[qq]+
2*Conjugate[pp].BB.qq.Inverse[II-AA].BB.qq.Conjugate[qq]+
Conjugate[pp].BB.Conjugate[qq].

Inverse[Exp[2*I*theta0]*II-AA].BB.qq.qq)]];
(*scaled leading coefficient \tilde{a}cr*)
a0=If[Im[mu0]==0,a00/Max[Power[Table[qq[[jj+m*n]],{m,0,r}],2]],
a00/(4*Max[Power[Abs[Table[qq[[jj+m*n]],{m,0,r}]],2]])];
StringJoin["Leading coefficient: ",ToString[a0]]

(*BIFURCATION DIAGRAM*)

plotopts={PlotRange->{{-1,1},{0,1}},AxesOrigin->{0,0},
PlotStyle->{Blue},Frame->True,FrameLabel->{"parameter","amplitude"},
FrameStyle->Directive[FontSize->16],PlotRangePadding->None,
AspectRatio->1,ImageSize->400};
Plot[{Sqrt[-s0*(par-parst)/a0]},{par,-1.5,1.5},Evaluate@plotopts]

